

VOL. 2
NO. 11

STATUS 1500

DECEMBER
1984

* 12 issues £10.50 (UK) * overseas £14.50 * published monthly *

* Edited by RONALD COHEN, 62 BLENHEIM CRESCENT, LONDON W.11 *



```
10 CLEAR :WAIT 1:DIM A$(0)*21,Z$(21)*1:A$(0)="YRREMSMAXYPHAHWENRAEY"
20 B=1:C=5:GOSUB 99:B=6:C=9:GOSUB 99:B=10:C=14:GOSUB 99:B=15:C=17:GOSUB 99
25 B=18:C=21:GOSUB 99:T$="YEAR":Q$="XMAS":R$="HAPPY":P$="MERRY":S$="NEW"
30 N=(U$=P$)+(V$=Q$)+(W$=R$)+(X$=S$)+(Y$=T$):D=(Z$(1)="M"):E=(Z$(10)="H")
35 FOR M=0 TO N+23*(N=5):BEEP 1,RND 64:NEXT M:L$=CHR$ 32
40 IF U$<>P$FOR J=1+D TO 5:R=D+RND (5-D):K$=Z$(J):Z$(J)=Z$(R):Z$(R)=K$:NEXT J
41 IF V$<>Q$FOR J=6 TO 9:R=5+RND 4:K$=Z$(J):Z$(J)=Z$(R):Z$(R)=K$:NEXT J
42 IF W$<>R$FOR J=10+D TO 14:R=9+E+RND (5-E):K$=Z$(J):Z$(J)=Z$(R):Z$(R)=K$:NEXT J
43 IF X$<>S$FOR J=15 TO 17:R=14+RND 3:K$=Z$(J):Z$(J)=Z$(R):Z$(R)=K$:NEXT J
44 IF Y$<>T$FOR J=18 TO 21:R=17+RND 4:K$=Z$(J):Z$(J)=Z$(R):Z$(R)=K$:NEXT J
51 FOR F=21 TO 25:@$(F)=CHR$ 0:NEXT F:FOR G=1 TO 5:U$=U$+Z$(G):NEXT G
53 FOR G=6 TO 9:V$=V$+Z$(G):NEXT G:FOR G=10 TO 14:W$=W$+Z$(G):NEXT G
55 FOR G=15 TO 17:X$=X$+Z$(G):NEXT G:FOR G=18 TO 21:Y$=Y$+Z$(G):NEXT G
57 PRINT U$;L$;V$;"-";W$;L$;X$;L$;Y$:GOTO 30
99 FOR A=B TO C:Z$(A)=MID$(A$(0),A,1):NEXT A:RETURN
```



98 MERRY XMAS!
99 SIGNALS - LETTER FROM DENMARK
100 SIGNALS - REPLY
101 PEEK POKE & MEMORY - XXI
102 POKER
104 SPACE LANDER

105 QUIZ
106 SKI-RUN
107 SKI-RUN the program
108 MINDBOGGLE CORNER
109 MORE MINDBOGGLE
110 MINEFIELD

112 SHEEP ON THE MOUNTAIN



Dear Reader - Thank you for your support over the past year. Whether you are one of those who have laboured to write long technical articles, or sent in programs or notes or queries, or just supported the newsletter with silent appreciation - whether you live in these islands, or far away across the globe, let me wish you

A VERY HAPPY XMAS AND NEW YEAR!



SIGNALS - LETTER FROM DENMARK

October 23, 1984

Mr. Ronald Cohen
62 Blenheim Crescent
London W. 11
England

Dear Sir,

I am very puzzled concerning your remark on page 79 of the October 1984 issue. Did you read my note concerning the newsletter, dated 22.8.84, or what ??

It is very sad that you seem to misunderstand me. Am I not expressing myself very clearly ? I did not mention anything about a CASIO TX500 which does not exist, but I did mention that I now own a CASIO FP-200. Maybe you should check your new printer or maybe the printer will choke on anything other than the SHARP PC-1500 ???

I am still using my PC-1500 daily. It is a beautiful machine. I like reading your newsletter which is an inspiration to me when developing new programs, although not programs for my PC-1500 owing to its small display.

My suggestion that you should consider expanding the newsletter was inspired by your remark on the front page of the June 1984 issue concerning the new subscribers. I think it sad if you are to discontinue the newsletter.

Have you read about the dinosaurs? They are all dead now. As I am informed they stopped evolving and were not able to adapt. I think that you are very concerned about solving today's problems with yesterday's computer technology only.

Do you think that all new computer buyers will buy the PC-1500 when the PC-1261 (10 K) and PC-1350 are now available ?


I am aware that new computers cost money and as I am not rich myself I have been forced to find some firms in Western Germany where I can buy my computers at a 30-40% discount compared to the Danish prices. The English prices are higher too.

I am sorry to disappoint you - AGAIN - to suggest that you might expand your newsletter which, I think, is of general interest to any computer fan.

I did not intend to step on a sore toe, nor do I wish you to write about my machine in particular.

I suggest that you read the American PORTABLE COMPUTER to see how much portable computers are evolving.

Sincerely yours,


Thorkild Rasmussen

8900 Randers
Denmark

30 october 84

Dear Mr Rasmussen

I am sorry I misquoted the code number of your latest new computer.

You are quite right in supposing that my printer cannot be driven by other machines that do not possess the necessary parallel interface.

I am gratified that you still find my newsletter useful. However I must emphasise that it is a newsletter for the PC 1500, and one of the main reasons for its success is my possession and use of the machine I am writing about. It would be quite easy to run a superficial magazine on portable computers in general, and this might attract new subscribers. Whether it would hold them might be another matter. Those who cannot remain constant to a machine are unlikely to remain constant to a publication.

But time and space would then prevent me from continuing to explore the PC 1500 in depth, which is the object of STATUS 1500.

I note your remark about dinosaurs. However the PC 1500 is evolving; the ME0/ME1 memory structure gives plenty of scope for further evolution; and while it continues to be in use I shall continue to explore its potential. When this is exhausted, perhaps I may consider writing about your latest toy, if indeed I can keep up with the vagaries of fashion.

You say I am trying to solve today's problems with yesterday's technology. Nothing of the sort! What I am trying to do is make the best use of the computer I have got, with the kind help of those subscribers to this newsletter who send information or programs or interesting queries about the PC 1500 for our mutual enlightenment, or amusement.

It is surely all a question of perspective. You look back on the dinosaurs with contempt, but you forget that they roamed this planet contentedly for a good few million years. Do you really think that pocket computers - or even pockets - will last as long? And when we become extinct, may it not well be because we develop new technology faster than we learn how to handle it?

yours sincerely

Ronald W. F. Cohen

PEEK POKE & MEMORY - XXI

A Ghost Program

It was midnight on Hallowe'en. The last innocent, whining "Trick or treat?" had been kicked down the front steps. All was quiet now. Idly, I switched on the computer. To my amazement, onto the screen flashed:

HELP!

I AM TRAPPED IN COMPUTER

I looked at the machine. I picked it up and shook it gently. Nothing rattled. I looked inside. Nothing there. I tried to switch to PRO mode to LIST the program. This was not easy, since somehow the computer was already in PRO mode. But the only LISTing found was a single line:

1: REM *THIS IS A GHOST PROGRAM

Baffled, I tried to RUN the program again. This time the message on the screen said:

////TOO LATE!

So I LISTed the program again. Now it had disappeared. Nothing at all. I promise, never again will I kick a child down the steps on Hallowe'en!

If you wish to try this program yourself, you will need to take a little care. Do not MERGE or mix it with another program. Make sure the length of each line is correct. Take care not to put an extra space after the REM in line 50. After you have experimented with the program, you will need to rehabilitate the program area by NEW 0.

```
10 ARUN :X=STATUS 2-STATUS 1:RESTORE 50:P=PEEK 30886:Q=PEEK 30887-3
20 BEEP 7:PAUSE "---HELP!":PAUSE "I AM TRAPPED IN COMPUTER"
30 POKE X+169,01:POKE X+170,78:POKE X+000,00:POKE X+000,00
40 POKE 30821,P,Q:POKE 30287,35:END
50 REM *THIS IS A GHOST PROGRAM
60 REM
70 PAUSE "////TOO LATE!":POKE 30823,PEEK 30821,PEEK 30822
```

STATUS 1 = 250

* * * * *

I am an idiot. At least, that is what I said to myself the other day. I had been driven almost to despair by a frustrating problem, and all the time the solution was staring me in the face. I was having cassette-recorder trouble. (Nothing to do with the machinery - the fault was traced eventually to an unstable cable connection). But meanwhile I had a very long program which it was essential to record. And every time I succeeded in CSAVEing part of the program, but not all. So why not do exactly that!

I divided the program into sections, and recorded each one by CSAVE M. I did each section twice, so that if there was a fault I could use the back-up. Since even this might fail, I located the beginnings of lines at suitable points, using PEEK 30886 and PEEK 30887 to find them: thus each section would start at the beginning of a line, making it easier to write in any lines totally missing. It took a little time, but nothing like as long as CSAVEing and CLOADing over and over again. In the event, it worked perfectly: I only had to use one back-up section. I also of course noted the pointers 30821 to 30826, so that I could re-insert them on reloading the program by sections.

* DRAW POKER *

by F.C.Odds

A pair, 2 pairs, 3 of a kind, a straight (consecutive numbers), a flush (all the same suit), a full house (3 of a kind and a pair), 4 of a kind, a straight flush. These are the winning hands at Poker. Can you beat the bank? The pay-off is evens on a pair (Jacks or better), 2 to 1 on 2 pairs; 3 to 1 on 3 of a kind, and so on.

Initiate by RUN. When your hand is displayed, you may draw further cards to improve your hand. Hold any card you wish to keep, by pressing the function key under that card. You may keep all 5 cards if you wish. Cancel the hold, if you change your mind, by the right-hand function key under 'hold?'. Then choose again. Press ENTER to fill your hand. The cards are not unlimited, but dealt from a pack of 52 cards, so it can help to remember what has been dealt. Play on until your winnings are lost, or keep them by staking \$0 in reply to the prompt.

```

10 RANDOM :WAIT 150:DIM A(4),B(4),C(4),D(4),A$(1)*30,B$(3),C$(13)*14
30 PRINT " *** DRAW POKER ***":R=1:P=0:S=100
50 B$(0)="000E1F3F7E3F1FOE":B$(1)="00185C7E7F7E5C18"
55 B$(2)="00081C4A774A1C08":B$(3)="00081C3E7F3E1C08"
60 C$(1)="7C1211127C0000":C$(2)="42615149460000":C$(3)="41494D4B310000"
62 C$(4)="1814127F100000":C$(5)="27454545390000":C$(6)="3C4A4949300000"
64 C$(7)="01017905030000":C$(8)="36494949360000":C$(9)="064949291E0000"
66 C$(10)="043E001C22221C":C$(11)="2040413F010000"
68 C$(12)="3E4151215E0000":C$(13)="7F081422410000"
70 A$(0)="00007F7F7F7F7F7F7F7F7F7F7F7F":A$(1)="0000000000000000000000000000
80 PRINT "You have $100 to stake":GOSUB 900
100 INPUT "Stake this deal? ";S1:IF S1=0GOTO 1600
102 S1=INT S1:IF (S1>S)+(S1=0)PAUSE "***IMPOSSIBLE!***":GOTO 100
105 WAIT :PRINT "Press <ENTER> for deal":P=0:CLS :WAIT 10
120 FOR K=0TO 4:D(K)=0:GDCURSOR (8+23*K):GPRINT A$(0);:NEXT K
130 GOSUB 1000:WAIT 2:GOSUB 1100
230 WAIT 0:CURSOR 21:PRINT "hold?":K=0
260 GDCURSOR (8+23*K):IF D(K)GPRINT A$(1);
270 A$=INKEY$ :IF A$<>" "GOTO 400
280 GDCURSOR (8+23*K):GPRINT C$(C(K));B$(B(K));
290 K=K+1:IF K=5LET K=0
300 GOTO 260
400 Y=ASC A$:BEEP 1,7*ABS (Y-16),300:IF (Y>22)+(Y<13)GOTO 260
410 IF Y=13GOTO 500
420 IF Y=22GOTO 450
430 IF Y>16LET D(Y-17)=1
440 GOTO 260
450 FOR K=0TO 4:D(K)=0:NEXT K:GOTO 260
500 WAIT 2:FOR K=0TO 4:IF D(K)GOTO 540
530 GDCURSOR (8+23*K):GPRINT A$(0);
540 NEXT K
550 GOSUB 1000:GOSUB 1100:E=1:F=0:G=0:H=0:WAIT 150
600 FOR K=0TO 4:IF B(1)<>B(K)LET E=0
620 NEXT K
630 F=C(0)*C(1)*C(2)*C(3)*C(4):G=C(0)+C(1)+C(2)+C(3)+C(4)
650 IF (F=17160)*(G=47)LET H=5
651 IF (F=720)*(G=20)LET H=1
652 IF (F=2520)*(G=25)LET H=1
653 IF (F=6720)*(G=30)LET H=1
654 IF (F=15120)*(G=35)LET H=1
655 IF (F=30240)*(G=40)LET H=1
656 IF (F=55440)*(G=45)LET H=1
657 IF (F=95040)*(G=50)LET H=1

```

00000


```

658 IF (F=154440)*(G=55)LET H=1
660 IF HGOTO 730
670 F=0:FOR K=0TO 3:FOR J=K+1TO 4
700 IF C(K)=C(J)LET F=F+1
710 NEXT J:NEXT K
730 IF E*HLET P=50*H:GOTO 800
740 IF ELET P=5:GOTO 800
750 IF HLET P=4:GOTO 800
760 IF F=6LET P=25:GOTO 800
770 IF F=4LET P=7:GOTO 800
780 IF F=3LET P=3:GOTO 800
790 IF F=2LET P=2:GOTO 800
795 IF F=1GOSUB 1200
800 FOR K=1TO 100:NEXT K
805 CLS :IF P=0GOTO 840
810 BEEP 2:P=P*S1:PRINT "You have won $";P:S=S+P:GOTO 850
840 S=S-S1
850 PRINT "You now have $";S
860 IF S=0GOTO 1500
885 IF R>42GOSUB 900
890 GOTO 100
900 POKE# 61447,67
910 Z=STATUS 2+2:PRINT "PACK BEING SHUFFLED..."
915 FOR K=1TO 52:POKE (Z+K),0:NEXT K
920 FOR K=1TO 52:BEEP RND 3,7,5
930 X=RND 52
940 IF PEEK (Z+X)GOTO 930
950 POKE (Z+X),K:NEXT K
970 R=1:POKE# 61447,0:RETURN
1000 FOR K=0TO 4
1010 IF D(K)GOTO 1050
1020 A(K)=PEEK (Z+R):R=R+1
1040 B(K)=INT ((A(K)-1)/13):C(K)=A(K)-13*B(K)
1050 NEXT K:RETURN
1100 FOR K=0TO 4:J=8+23*K
1110 IF D(K)GOTO 1130
1120 BEEP 1,J,(1000/J+10)
1130 GCURSOR J:GPRINT C$(C(K));B$(B(K));
1140 J=J+23:NEXT K:RETURN
1200 FOR K=0TO 3
1210 FOR J=K+1TO 4
1220 IF (C(K)=C(J))*(C(K)>10)LET P=1
1225 IF (C(K)=C(J))*(C(K)=1)LET P=1
1230 NEXT J:NEXT K:RETURN
1500 PRINT "You are broke..."
1510 PRINT "tough!!":GOTO 1610
1600 IF S>100PRINT "You came out ahead!":WAIT 200:PRINT "You won $";S-100:
1610 S4=100-S:PRINT "You lost $";S4
1620 PRINT "It could have been worse!"
1630 IF PEEK 40960<>255GOTO 4000+2000*SGN (S-100)
1640 END
2000 LPRINT "you lost $";S4:END
4000 LPRINT "you broke even!":END
6000 LPRINT "you won $";S-100:END

```

GOTO 1630

STATUS 1 = 2685 approx.

SPACE LANDER

by Tony Willman

```
10 "A"CLS :WAIT 150:PRINT "SPACE LANDER by T Willman"
15 PRINT "Fuel & oxygen are limited":BEEP 1
20 PRINT "To burn fuel, press SPACE":BEEP 1
30 REM SET UP VALUES
35 CLEAR :A=5:L=3:B=1:B$="63223E361C08":C$="6B2A3E361C08"
36 D$="7C6B7752776B7C"
40 D=775:F=44-5*B:V=5-RND (2*B):O=44-5*B:WAIT 0:RANDOM
50 REM SET UP SCREEN
60 CLS :GCURSOR 42:GPRINT "724565454200552A557F";
65 FOR X=0TO 45-5*B:GCURSOR X
70 GPRINT "7F";:NEXT X
80 REM MAIN ROUTINE
90 A$=INKEY$
100 IF A$=" "AND F>0AND O>0GOTO 500
110 V=V+1:Z=2
120 E=D:D=D-V
130 IF D<265LET D=264:IF V>ALET Z=4
140 E=INT (E/5):IF E>155LET E=155
150 GCURSOR F:GPRINT (&4FAND POINT F)
160 GCURSOR O:GPRINT (&79AND POINT O)
170 GCURSOR E:GPRINT "0000000000000";:IF D>775GOTO 190
180 GCURSOR INT (D/5):GPRINT @$ (Z);
190 O=O-0.2*(O>0):IF D>=265BEEP 1,1+5*F,2:GOTO 90
200 REM LAND ROUTINE
205 IF V>AOR O<=0GOTO 300
210 BEEP 10,20,200:WAIT 150:CURSOR 0:PRINT "Well done, you have landed"
220 A=A-1:IF A>2-(B>3)PRINT " It is now harder!":S=S+20*(4-A)-F:GOTO 40
230 REM END OF STAGE ROUTINE
240 PRINT "Congratulations. You have":BEEP 5
250 PRINT "completed stage No.":B
260 A=6-B:B=B+1:IF B<5LET L=L+1:GOTO 40
270 REM GALACTIC CHAMPION ROUTINE
280 PRINT " You are now a....":BEEP 2
290 PRINT "***GALACTIC CHAMPION***":BEEP 5:GOTO 350
300 REM LOSING ROUTINE
310 BEEP 10,200,50:WAIT 150:CURSOR 0:PRINT " Hard luck, you have"
320 IF O<=0PRINT " run out of oxygen, and.."
330 PRINT " ...CRASHED...":BEEP 1,217,483
340 L=L-1:IF L>0GOTO 40
350 PRINT "You scored";INT S;" points"
360 END
500 V=V-1:Z=3:F=F-0.5*(F>0):BEEP 1,255/(ABS V+1),90:GOTO 120
```

STATUS 1 = 1286

QUIZ

- | | |
|--------------------------------------------------------------------------------------------------------------------------|---------------------|
| 1) So what <u>does</u> it say on the back of your PC 1500?
You know the words by now -
but what about the numbers? | Look and see. |
| 2) What will you find in PEEK 30819? | See page 61 |
| 3) What is the simplest sort of sort? | page 65. |
| 4) What do 30900, 30901, 30898, 30899
and 30875 have in common? | page 41 |
| 5) What do Gerhardt and Cornelia do on Sundays? | page 70 |
| 6) What does SHIFT/CLEAR do? | page 3 |
| 7) What happened to IAN TRAYNOR's cat last Xmas? | page 2 |
| 8) Why does the editor get so cross
when you use the variables I and O? | !!!! |
| 9) What are between 31200 and 31208? | page 4 |
| 10) Is it best to turn your computer off,
or let it turn itself off? | page 67 |
| 11) Why is BASIC misleading? | page 91 |
| 12) What can you do with CHR\$ 162? | page 79 |
| 14) A01, A03, A04 - which is which? | page 18 |
| 13) Distinguish between GOTO "A
and GOTO "A | page 40 |
| 15) 1 - 3 - 5 - 7 - St. Andrew or St. George? | page 72 (diagram 1) |
| 16) INPUT "prompt?";A
or INPUT "prompt?";A
- does it matter? | page 45 |
| 17) How do you get ERROR 28? | page 71 |
| 18) And ERROR 15? | page 65 |
| 19) What was unjustified? | page 22 |
| 20) What was "the best excuse yet?" | page 84 |

SCORE: 5 for each correct answer.
 4 for each wrong answer.
 3 for "don't know". (After all, its Xmas!)

SCORE CHECK:	98 - 100	Sure you did not cheat?
	87 - 95	Fantastic!
	78 - 84	You are too impetuous.
	60 - 76	Average reader.
	below 60	Add up your score again.

ABSOLUTELY NO PRIZES!

SKI-RUN

Tracks in the snow through the pinewoods. Trudge back to the start for another downhill run. The slopes are littered with fallen skiers, and pocket computers. The track is 500 in length.

Rules: your run is disqualified if a) you hit a tree or b) go over a precipice at the edge or c) miss the finishing line or d) hit a marker post or e) start before the 'GO' signal or f) cross the line so fast that you hit the spectators in the stand behind the line.

Controls: Initiate by RUN.
DEF N gives a new course.
DEF A gives another run on the same course.
DEF C restores the "READY?(=)" prompt.
DEF G gives 'reverse linefeed'.

Operation: There are 2 options, Easy(E) and Hard(H). 'Hard' gives on average shorter start and finish lines, more trees, larger trees, more obtrusive marker posts. When the run has been delineated, the pen returns to near the start line. The prompt 'READY?(=)' appears. Use the 4 cursor keys to position the pen exactly on the START line, as desired. When equal to the fray, key '='. Enter 'speed' and 'acceleration' For novices, speed 10 and acceleration 1 are recommended. Wait 10 seconds. Then you will see - 5 - 2 - 1 - GO.

Use the 6 function keys to move. You see on the display something like:

1 10 0.0001 1

The first '1' indicates curvature to right. '-' for curve left.
'10' indicates speed.
'0.0001' is elapsed time in seconds.
'1' on the far right indicates rate of acceleration.

The cross-movement is not measured by angle, but by absolute movement across. This amount increases, each move. The right-hand function key adds 5 each time to the X value. (X value is the amount across). The left-hand function key subtracts 5. The innermost keys add and subtract 1 respectively. The intermediate keys add or subtract 3. Remember, the 3 right-hand keys ADD, for movement to the right; the 3 left-hand keys subtract, for movement to the left. BUT if you are swerving sharply to the right the opposite key will not immediately change direction: it will first gradually correct to ahead, before going the other way. Remember also the proportion of curve depends on speed: the higher the speed, the less the angle.

While running, you may alter acceleration value by number keys 1 to 9. You may DECELERATE by keying '-', and the acceleration value will become a deceleration value. Restore to acceleration by '+' key. In emergency number key '0' will slither to a halt, allowing speed and acceleration to be reset; but there will be a slight delay before you can move off again. At the end of each run, key ENTER to return to starting area. It sounds more complex than it is, but the operational controls are quickly learnt. Nevertheless smooth fast runs are by no means easy.

Happy skiing!

SKI-RUN - the program

```

1 "N"CLEAR :TEXT :GRAPH :H=0:E=1:GLCURSOR (0,-800):SORGN
7 INPUT "easy(E)or hard(H)?" :Q:BEEP 1+Q,30,60
10 A=RND 210:B=RND 210:C=RND 210:D=RND 210
15 IF ABS (A-B)<(20*Q+49)OR ABS (C-D)<(20*Q+49)GOTO 10
19 CSIZE 5:LINE (A,0)-(B,3),0,3,8
20 LINE (C,500)-(D,500),0,3:POKE 31204,0:LINE (0,590)-(210,610),1,3,B
21 GLCURSOR (C,480+10*Q):LPRINT "#":GLCURSOR (D,480+10*Q):LPRINT "#"
22 GLCURSOR (A,-10):LPRINT "#":GLCURSOR (B,-10):LPRINT "#"
30 COLOR 2:FOR F=1TO 11+RND 10-6*Q:CSIZE 4+RND 5-4*Q
50 GLCURSOR (RND 210,30+30*Q+RND (410-30*Q)):LPRINT "":NEXT F
900 GLCURSOR (0,0):GOSUB 8030
910 TEXT :GRAPH
920 "C"GLCURSOR (X,Y):ON ERROR GOTO 910
1000 BEEP 1,15,30:WAIT 0:PRINT "ready?(=)":GOTO 1000+ASC INKEY$
1008 X=X-1:GOTO 920
1010 Y=Y-1:GOTO 920
1011 Y=Y+1:GOTO 920
1012 X=X+1:GOTO 920
1061 "=":X=0:Y=5:T=0:SORGN :POKE 31204,0,1:GOTO 1990
1990 GOSUB 8070:USING :TIME =0:WAIT 0:K=ASC INKEY$ :T=0
2005 IF TIME <.0010BEEP 1,1,1:PRINT ".":GOTO 2005
2010 BEEP 1,70,200:CURSOR 9:PRINT "5":TIME =0
2025 IF TIME <.0003GOTO 2025
2030 BEEP 1:CURSOR 9:PRINT "2"
2035 IF TIME <.0004GOTO 2035
2040 BEEP 1:CURSOR 9:PRINT "1"
2045 IF TIME <.0005GOTO 2045
2048 K=ASC INKEY$ :IF K<>0STOP
2090 BEEP 1:CURSOR 9:PRINT "GO":TIME =0:X=0:Y=5:F=1:IF V=0LET V=1
2995 ON ERROR GOTO 3000:WAIT 0
3000 PRINT X;Y;TIME ;" ";V*F:GOTO 3000+ASC INKEY$
3017 X=X-5:GOTO 3100
3018 X=X-3:GOTO 3100
3019 X=X-1:GOTO 3100
3020 X=X+1:GOTO 3100
3021 X=X+3:GOTO 3100
3022 X=X+5:GOTO 3100
3043 F=1:BEEP 1,25,95:GOTO 3000
3045 F=-1:BEEP 2,25,95:GOTO 3000
3048 BEEP 9,25,20:RLINE -(X/2,Y/2),1:GOSUB 8070:X=0:Y=5:F=1:GOTO 3000
3049 V=1:GOTO 3099
3050 V=2:GOTO 3099
3051 V=3:GOTO 3099
3052 V=4:GOTO 3099
3053 V=5:GOTO 3099
3054 V=6:GOTO 3099
3055 V=7:GOTO 3099
3056 V=8:GOTO 3099
3057 V=9:GOTO 3099
3099 BEEP 2,10,90:GOTO 3000
3100 Y=Y+V*F:IF ABS Y<1LET Y=SGN Y
3120 RLINE -(X,Y),L:T=T+Y:POKE 31204,0,1
3160 IF T>505WAIT :TM=TIME :USING "###.###":PRINT TIME :USING :GOTO "A"
3180 GOTO 3000
5000 "A"CLS :SORGN :GLCURSOR (0,-T):X=0:Y=0:T=0:F=0:V=0:SORGN
5001 POKE 31204,0,1:N=N+1
5003 INPUT "clear run?(*/0)";P$:IF P$<>"*"LET P$=CHR$ 32
5004 CSIZE 2:W=X-PEEK 31206
5005 GLCURSOR (W,-25*N):GOSUB 8090:LPRINT TM$;P$;CHR$ (82-2*(L=7))
5006 GLCURSOR (W,-100-25*N):BEEP 12,12,60:GLCURSOR (W,-20):POKE 31204,0,1
5010 TEXT :GRAPH :GOSUB 8030:USING :TIME =0:GOTO 920
7030 "G"POKE 31204,0,1:TEXT :LF -8:GRAPH :END
8030 R=0:P=7:INPUT "color";C:COLOR C:INPUT "practise(P)or RACE(R)?" :L:RETURN
8070 INPUT "speed? ";S:INPUT "acceleration? ";V:RETURN
8090 TM$=STR$ TM
8091 IF LEN TM$<6LET TM$=TM$+"0":GOTO 8091
8092 RETURN

```

STATUS 1= 1809

MINDBOGGLE CORNER

Our Xmas competition is for a self-portrait, using the CE 150. You could use the sketch program in the TANDY Games Cassette, or our own SUPERSKETCH (vol.1, page 99), or any program you prefer. Not in color, please. We would very much like to see what you all look like - or what you think you look like! We would particularly welcome entries from all those whose names have appeared in the newsletter over the past year. Portraits may be realistic, or symbolic, as you prefer.

* Deadline: 15th January 1985.

Thank you for responding so nobly to my plea for entries for the 5-liner competition. At the last minute a splendid collection of games and entertainments was received. C.P.UNDERWOOD sent 2 entries, while JOHN WARNER sent no less than 4! He narrowly missed the prize, but expressions such as IF...THEN GOTO... were thought to be so contrary to the spirit of economy which inspired the competition that - regretfully - he was relegated to 2nd place. Such expressions may be good grammar but they are appalling BASIC. DAVID RIHOY originally sent his entry with 3 pages of rules and explanations: he has condensed these down to a few lines, and I have difficulty playing the game. You may be brighter than I am. NUMBER? by L.E.SIMONS was a strong contender, and I am slightly ashamed of my preference for the less scientific RANDOM RANDOM by C.P.UNDERWOOD, despite the random explanations which accompany it.

The clear winner is GOLF by TONY WILLMAN. The program is entirely self-contained and self-explanatory, and has a fair balance of skill and chance. Below and overleaf are printed a selection from the games and entertainments sent in.

POT THE PIGS & their revenge

by John Warner

Display indicates current pig number, shots remaining for current pig, pigs shot. Fire by keying *. Don't let them get away!

```
1 P=0:FOR Y=1TO 9:FOR Z=36TO 132STEP 2:IF Y=9GOCURSOR Z:PRINT "-":BEEP 1,9,9:GOCURSOR Z+1:PRINT " " :NEXT Z:WAIT :PRINT :END
2 A$=INKEY$ :GOCURSOR 135:GPRINT "0818507B1F3C78602040":ON ERROR GOTO 2:W=5:FOR X=120TO 25STEP -1:WAIT W
3 CURSOR 24:PRINT P:IF INKEY$ = "*"AND W=1THEN GOSUB 5:CURSOR 0:PRINT W;Y:IF T=X+11THEN GOCURSOR X:WAIT 25:PRINT "":P=P+1:WAIT 0:NEXT Y
4 GOCURSOR X:GPRINT "06060D0F3E5E1E1E5E3C04":GOCURSOR X-1:GPRINT "06060D4F3E1E1E1E7C040":NEXT X:NEXT Y:END
5 RANDOM :WAIT 0:BEEP 1,25,50:FOR T=123TO X+9+RND 8STEP -1:GOCURSOR T:PRINT "- ":NEXT T:GOCURSOR T:PRINT "":W=W-1:RETURN
```

NUMBERS GAME

by L.E.Simons

Choose maximum from which computer picks random number. Then guess number computer has picked. Guessing halfway ("binary chop") is quickest method. Where max.<2N, approximately N guesses should suffice. Finally, the mean and variance indicate closeness of guessing.

```
5000 "G"RANDOM :CLS :A$="Small":B$="Large":WAIT 0:K=0:S=0:S2=0:N=1:INPUT "Max: ";M:BEEP 1:V=RND
M
5010 "G1"INPUT "Guess: ";G:BEEP 1:K=K+1:IF G<>VWAIT :PRINT "Too ";@$(1+(G>V)):WAIT 0:GOTO "G1"
5020 WAIT :PRINT USING " "; "Right in";K:N=N+1:S=S+K:S2=S2+K*K:K=0:R$="Y":INPUT "Again? ";R$
5030 IF R$<>"N"LET V=RND M:GOTO "G1"
5040 PRINT USING "####.###"; "Mean";S/N:PRINT "Variance: ";(S2-(S*S/N))/(N-1):END
```


WORD SQUARE

by C.P. Underwood

Computer writes 9*9 square of random letters. Join adjacent letters (horizontally, vertically, diagonally) to make words. Do not cross lines, do not re-use letters. Twist and zigzag as required. Score 1 per letter for 2 and 3-letter words; 2 per letter for 4-letter words; 3 per letter for 5-letter words, and so on. Time limit: 3 minutes.

```

6 CLEAR :TEXT :RANDOM :CSIZE 2:DIM W$(9,9)*1:FOR X=1TO 9:FOR Y=1TO 9:BEEP 1,1:W$(X,Y)=CHR$ (
64+RND 26):NEXT Y:NEXT X
7 "C"FOR X=1TO 9:FOR Y=1TO 9:LPRINT W$(X,Y);CHR$ 32;:NEXT Y:LF 1:NEXT X:LF 7
8 PAUSE "tear off -":PAUSE "another copy - DEF C":PAUSE "start by DEF Z":END
9 "2"TIME =0:WAIT 0:USING "###.###"
10 CURSOR 10:PRINT 100*(TIME <.03)*TIME :BEEP (TIME >.03),90,90:GOTO 10

```

GOLF

by Tony Willman

```

1 CLEAR :WAIT 150:PRINT "GOLF by T. Willman":FOR H=1TO 18:D=200+RND 200:P=INT (D/100):S=0
2 BEEP 1:PRINT "Dist.":D;" /Par":P;" Shots=":S:INPUT "Club? ":C:S=S+1
3 SS=C*14+RND (C*5):D=ABS (D-SS):IF D>20THEN 2
4 BEEP 5,10:PRINT "Hole":H;" in":S;" shots":ST=ST+S:PT=PT+P:NEXT H
5 PRINT "You went round in":ST;" shots":PRINT "You are":ST-PT;" over par"

```

An alternative line 3 may make the game trickier:

```

3: SS=C*7+C*7*(C>2)+RND (C*5): D=ABS (D-SS): IF D>10*(C>3)+RND 13 THEN 2

```

MEMORY GAME

by David Rihoy

1-4 players. RUN to start. Enter a number 1-26 in answer to prompts "1st" and "2nd". If the 2 LETTERS printed are the same, 1 point is scored. Key ENTER for score, DEF G for next player. Game restarts when all letters correctly matched.

```

5 CLEAR :T=26:WAIT :DIM X$(1,T)*1:FOR J=1TO T:X$(0,J)=CHR$ (64+J):X$(1,J)=X$(0,J):NEXT J:FOR
I=0TO 1
10 FOR J=TTO 1STEP -1:BEEP 1,1:K=RND J:B$=X$(I,K):X$(I,K)=X$(I,J):X$(I,J)=B$:NEXT J:NEXT I:BE
EP 3
15 "3"INPUT "Player?":P:INPUT "1st?":X:INPUT "2nd?":Y:A$=X$(0,X):B$=X$(1,Y):PRINT A$,B$:Z=(A$=
B$)
20 C$="+":Z=Z*(A$<>C$):@ (P)=@ (P)+1:@ (P+4)=@ (P+4)+Z:IF ZLET X$(0,X)=C$:X$(1,Y)=C$
25 S=S+Z:PRINT "P1.":P;" has won":@ (P+4);" in":@ (P);" goes":GOTO (15-10*(S=T))

```

RANDOM RANDOM

by C.P. Underwood

Computer picks random number <1000, and displays a random number of random numbers each less than original random number. You are only allowed a random number of random guesses to discover the RANDOM NUMBER.

```

1 RANDOM :BEEP 5:CLEAR :N=RND 1000:R=INT (5+RND (N/40)):PAUSE " - new number!"
2 W=45/(C+1):WAIT W:FOR I=1TO RND 7:PRINT RND N;:NEXT I:PRINT :INPUT G:C=C+1:GOTO 4-3*(C=R)+
(C<R)*SGN (G-N)
3 BEEP 1,99:PRINT G;" is under - #":STR$ C;" goes":GOTO 2
4 WAIT 250:BEEP 1,100:BEEP 1,90:PRINT G;" is correct in #":STR$ C;" goes":GOTO 1
5 BEEP 1,77:PRINT G;" is over - #":STR$ C;" goes":GOTO 2

```

CHEAT

by John Warner

Match numbers displayed by keying *. Outer numbers on screen indicate your score, computer's score. Inner numbers displayed are to be matched. Keying * when no match displays 'CHEAT'.

```

1 DIM A(41):X=0:FOR Y=1TO 4:FOR Z=1TO 10:X=X+1:A(X)=Z:NEXT Z:NEXT Y:P=50:Q=P:B=70:S$="CHEAT"
2 FOR Z=1TO 40:Y=RND 40:A(0)=A(Y):A(Y)=A(Z):A(Z)=A(0):NEXT Z:FOR X=1TO 40:IF B$=""PRINT S$
3 WAIT B:PRINT P;" ";A(X);" ";A(X+1);" ";Q:B=B-1:IF A(X)=A(X+1)THEN LET A$=INKEY$:GO
SUB 5
4 B$=INKEY$:NEXT X:ON ERROR GOTO 3:B=B+35:RANDOM :GOTO 2
5 WAIT B/2:IF A$=""THEN PRINT "SNAP":BEEP 2,99,99:P=P+2:Q=Q-1:RETURN

```

H.M.S.STATUS (a frigate of the 1500 class) is steaming at full speed to the relief of Mafeking. An unscrupulous enemy has mined the Baltic seas. However the frigate is equipped with the latest mine-detection device, purchased from USA at the cost of scrapping the entire National Health Service. This emits a high-pitched tone when within 2 squares of a mine, and an even more piercing tone 1 square away. But NOT diagonally, only vertically and horizontally. This is because the ship is equipped with the latest technological advance in steering-gear (cost: 3 universities and the Brigade of Guards) and cannot navigate diagonally. Enter at bottom of 10*10 grid, exit only at the top. Steer by the 4 number keys - 2 - 4 - 8 - 6. How quickly can you get through the minefield? There is 1 mine per row: at Skill Level 2 there may be more than 1 mine on each row, and some of them could even explode without warning.

```

10 CLEAR :WAIT 0
20 POKE &7750,90,10,106,1,72,1,74,1,190,230,111,96,110,128,153,8
25 POKE &7760,82,153,17,154,90,1,190,119,82,154
30 CURSOR 7:PRINT "MINEFIELD!!";
40 CALL &7750
50 DIM A$(5)*12,B$(5)*12,C$(5)*12,D$(6)*12,E$(6)*12,F$(1)*60
60 A$(0)="7F4141414141":A$(1)="7F0202020202":A$(2)="7F0404040404"
62 A$(3)="7F0808080808":A$(4)="7F1010101010":A$(5)="7F2020202020"
64 B$(0)="7F4155495541":B$(1)="7F02160A1602":B$(2)="7F04140C1404"
66 B$(3)="7F081C081C08":B$(4)="7F1014181410":B$(5)="7F2034283420"
68 C$(0)="7F4155495541":C$(1)="41415549557F":C$(2)="414155497F41"
70 C$(3)="4141557F5541":C$(4)="41417F495541":C$(5)="417F55495541"
72 D$(1)="7E0202020202":D$(2)="7C0404040404":D$(3)="780808080808"
74 D$(4)="701010101010":D$(5)="602020202020":D$(6)="404040404040"
76 E$(1)="7E02160A1602":E$(2)="7C04140C1404":E$(3)="78081C081C08"
78 E$(4)="701014181410":E$(5)="602034283420":E$(6)="404054485440"
79 E$(0)="7F7F7F7F7F7F"
80 "A"CLS :INPUT "HOW MANY PLAYERS? ";S:U=0:RR=0
90 S=ABS INT S:IF S=0GOTO 80
95 INPUT "Skill level? (1/2)";SK:SK=SK-1
96 IF SK<0OR SK>1GOTO 95
100 FOR Z=1TO S:WAIT 0:RANDOM
110 FOR X=3TO 12:@(X)=RND 10:NEXT X
120 Q=3:PRINT "PLAYER ";Z;:BEEP 8:CLS
130 INPUT "Start square (1-10)? ";P:CLS
140 IF P<1OR P>10OR P<>INT PGOTO 130
150 TIME =0
160 "D"GOSUB 900
170 F$=A$(0):G$=B$(0)
180 GOSUB 950
190 GOSUB 780
280 CURSOR 0:PRINT "Row";Q-2
282 GOTO "C
285 X=1
290 A$=INKEY$ :IF A$<>"GOTO "M
300 X=X+1:IF X<10GOSUB 780:GOTO 290
310 GCURSOR 6*P+44:GPRINT A$(0);
320 X=1
330 A$=INKEY$ :IF A$<>"GOTO "M
340 X=X+1:IF X<5GOSUB 780:GOTO 330
350 GCURSOR 6*P+44:GPRINT B$(0);
360 GOTO 285
400 "M"R=ASC INKEY$ :IF R<50OR R>56GOTO 330
410 R=R-48:IF R/2<>INT (R/2)GOTO 285
420 ON R/2GOTO "B","L","R","F"
430 "F"IF Q-1>12GOTO "E
440 FOR Y=1TO 5:GOSUB 900
450 F$=A$(Y):G$=B$(Y)
460 GOSUB 950:GOSUB 780
470 NEXT Y

```

```

480 Q=Q+1:GOTO "D
490 "B"IF Q-1<3BEEP 1:GOTO 285
500 FOR Y=5TO 1STEP -1:GOSUB 900
510 F$=A$(Y):G$=B$(Y)
520 GOSUB 950:GOSUB 780
530 NEXT Y
540 Q=Q-1:GOTO "D
550 "L"IF P-1=0BEEP 1:GOTO 285
560 FOR Y=5TO 1STEP -1
570 GCURSOR P*6+38+Y:GPRINT C$(Y);
580 FOR X=1TO 20:NEXT X:GOSUB 780
585 NEXT Y
590 P=P-1:GOTO "D
600 "R"IF P+1=11BEEP 1:GOTO 285
610 FOR Y=1TO 5
620 GCURSOR P*6+44+Y:GPRINT C$(Y);
630 FOR X=1TO 20:NEXT X:GOSUB 780
640 NEXT Y
650 P=P+1:GOTO "D
660 "E"TIME=TIME*100:GOSUB 780
670 FOR Y=1TO 5:GOSUB 900
680 F$=D$(Y):G$=E$(Y)
690 GOSUB 950
700 NEXT Y
710 F$=D$(6):G$=E$(6):GOSUB 900:GOSUB 950
720 POKE# &F007,0:BEEP 7:CLS
730 PAUSE "You did it!!"
740 PAUSE "It took you ";T;" minutes"
750 IF (T<RR)+(RR=0)LET RR=T:U=Z
755 PAUSE "best time = ";RR
760 WAIT :PRINT "KEY <ENTER> TO CONTINUE"
770 GOTO 860
780 GCURSOR 115:PRINT USING "###.##";100*TIME ;:USING :RETURN
800 "X"FOR X=1TO 6
810 GCURSOR 6*P+44:GPRINT E$(0);
820 CALL &7764
830 GCURSOR 6*P+44:GPRINT B$(0);
835 CALL &7764
840 NEXT X:CLS :POKE# &F007,0
850 PAUSE "You hit a mine - SUNK!"
860 NEXT Z:IF U=0GOTO 880
870 WAIT :PRINT "Player ";U;" wins in";RR
880 END
900 F$(0)="" :F$(1)="" :RETURN
910 "C"X=0:IF @(Q)=PGOTO "X
915 IF SK=1AND @(Q+1)=PGOTO "X
920 IF @(Q+2)=POR @(Q-2)=POR @(Q)=P-2OR @(Q)=P+2LET X=68
925 IF SK=1AND (@(Q+3)=POR @(Q-1)=POR @(Q+1)=P-2OR @(Q+1)=P+2)LET X=68
930 IF @(Q+1)=POR @(Q-1)=POR @(Q)=P-1OR @(Q)=P+1LET X=66
935 IF SK=1AND (@(Q+2)=POR @(Q+1)=P-1OR @(Q+1)=P+1)LET X=66
940 POKE# &F007,X:GOTO 285
950 FOR X=1TO 5:IF X=PGOTO 960
955 F$(0)=F$(0)+F$:GOTO 965
960 F$(0)=F$(0)+C$
965 NEXT X
970 FOR X=6TO 10:IF X=PGOTO 980
975 F$(1)=F$(1)+F$:GOTO 985
980 F$(1)=F$(1)+G$
985 NEXT X
990 GCURSOR 50:GPRINT F$(0);F$(1);"7F";
995 RETURN

```

STATUS 1 = 2877

SHEEP

by ?????

"The mountain sheep are sweeter, but the valley sheep are fatter!"
 The thin sheep are coming down the mountain; the fat sheep are going up.
 The track is so narrow that they cannot pass. They must jump over each other.
 A sheep can jump over only one opposing sheep at a time, and must have a space to jump into. Eventually, they will have changed places.
 Execute by the extreme left and extreme right 'Function Keys'.
 It may seem impossible, but IT CAN BE DONE!

```

4 DIM A$(2)*24:Z=0:A$(0)="000000000000000000000000"
6 A$(1)="017E7E1E1E1E1E7E7E030707"
8 A$(2)="0707037E7E0E0E0E0E7E7E01"
10 Z=Z+1:A=2:B=2:C=2:D=2:E=0:F=1:G=1:H=1:I=1
12 M=0:WAIT 0:FOR L=1TO 9
14 IF @(L)=2GPRINT A$(1)
16 IF @(L)=1GPRINT A$(2)
18 IF M=24THEN 24
20 M=M+3:CUSOR M
22 NEXT L
24 A$=INKEY$ :BEEP (ASC A$=17),17,199:BEEP (ASC A$=22),255,17
26 IF (A=1)*(B=1)*(C=1)*(D=1)*(E=0)*(F=2)*(G=2)*(H=2)*(I=2)GOTO 56
28 IF A$=CHR$ 17GOTO 34
30 IF A$=CHR$ 22GOTO 42
32 GOTO 24
34 FOR P=1TO 8
36 IF (@(P)=2)*(@(P+1)=0)GOTO 90
38 IF (@(P)=2)*(@(P+1)=1)*(@(P+2)=0)GOTO 64
40 NEXT P:GOTO 50
42 FOR Q=1TO 8
44 IF (@(Q)=0)*(@(Q+1)=1)GOTO 70
46 IF (@(Q)=0)*(@(Q+1)=2)*(@(Q+2)=1)GOTO 80
48 NEXT Q
50 WAIT 0:IF Z<3CLS :PRINT "BLOCKED! TRY AGAIN":GOSUB 98:GOTO 10
52 WAIT 150:IF Z<5PRINT "TRY AGAIN!":GOTO 10
54 PRINT "TRY HARDER ":GOTO 10
56 WAIT 0:IF Z=1PRINT "TERRIFIC-HOME IN 1 !":GCURSOR 92:GOTO 100
58 IF Z<4WAIT 0:PRINT "BRAVO*WIN IN ";Z;" GOES":GCURSOR 74:GOTO 100
60 WAIT :IF Z<6PRINT "NOT BAD - ";Z;" GOES":END
62 PRINT "NULL ! ! ! ";Z;"GOES":END
64 @(P)=0:@(P+2)=2:M=3*P-3
66 CURSOR M:GPRINT A$(0):IF M<19CURSOR M+6:GPRINT A$(1)
68 GOTO 86
70 @(Q)=1:@(Q+1)=0:M=3*Q-3
72 CURSOR M:GPRINT A$(2)
74 CURSOR M+3:GPRINT A$(0)
76 GOTO 86
80 @(Q)=1:@(Q+2)=0:M=3*Q-3
82 IF M<19CURSOR M+6:GPRINT A$(0)
84 CURSOR M:GPRINT A$(2)
86 IF (A$=CHR$ 17)+(A$=CHR$ 22)LET A$="V":A$=INKEY$ :GOTO 86
88 GOTO 24
90 @(P)=0:@(P+1)=2:M=3*P-3
92 CURSOR M:GPRINT A$(0)
94 CURSOR M+3:GPRINT A$(1)
96 GOTO 86
98 GCURSOR 14:GPRINT "0B07":GCURSOR 94:GPRINT "5655"
99 GCURSOR 85:WAIT 150:GPRINT "4546":RETURN
100 WAIT :GPRINT "5655":END

```

STATUS 1 = 1265